Лекция 13. Процедуры и функции


В программах часто необходимо для решения основной задачи выполнять одну и ту же подзадачу несколько раз, с разными аргументами. Например, если надо найти наибольший общий делитель четырех чисел, то его можно найти через наибольший общий делитель двух чисел, т. е. НОД4(a,b,c,d)=НОД(a,НОД(b,НОД(c, d))), то есть задачу можно свести к нахождению НОД двух чисел, выполненному 3 раза. Чтобы не выписывать в программе один и тот же алгоритм решения подзадачи несколько раз, используют процедуры (называемые также подпрограммами).
Определение 13.1. Процедура (подпрограмма) – именованное, специальным образом оформленное описание решения подзадачи, записанное вне раздела операторов программы. 
<описание процедуры>::= <заголовок процедуры> ; (<блок> | <директива>) | 
procedure <имя> ; <блок>
<заголовок процедуры> ::= procedure <имя> [( <список формальных параметров> )]

Всякий раз, когда требуется обратиться к процедуре, в основной программе помещают оператор вызова процедуры:

<оператор вызова процедуры>::= <имя процедуры> [( <список фактических параметров> )]

Рассмотрим пример:
program Пример (input, output);

var x,y:integer;
Procedure digiSum; { поиск суммы цифр целого числа (y:= сумма цифр x) }

var temp:integer;
begin
у:=0; temp:=abs(x);

while (temp>0) do begin


y:=y + temp mod 10; temp:=temp div 10;

end;
end;
begin
read(x); digiSum; x:=y; digiSum; writeln(y)

end.

Программа ищет сумму цифр суммы цифр числа x. Выигрыш от использования процедуры очевиден (не надо 2 раза выписывать цикл). Однако пользоваться процедурой digiSum неудобно, каждый раз перед вызовом надо заносить значение в x. Очевидно, что число, для которого процедура ищет сумму цифр, является её параметром. Изменим программу:
…
Procedure digiSum1(i:integer); { поиск суммы цифр целого числа (y:= сумма цифр i) }

begin
у:=0; i:=abs(i);

while (i>0) do begin


y:=y + i mod 10; i:=i div 10;

end;
end;
begin
read(x); digiSum1(x); digiSum1(y); writeln(y)

end.

i называется формальными параметрами процедуры и позволяет настраивать её при каждом вызове на работу с другим целочисленным значением. Теперь, вызывая процедуру, мы указываем в операторе вызова фактический параметр ​– те значение, с которым должна работать процедура. Если параметров несколько, то при вызове первый фактический параметр подставляется на место первого формального, второй фактический – на место второго формального и т.д. В нашей процедуре использован параметр-значение, поскольку его роль передать входное значение в процедуру. Рассмотрим как происходит передача значения при первом вызове процедуры: digiSum1(x)
Все происходит так как если бы в блоке процедуры было написано:

var i:integer;

begin i:=x;  …
То есть, заводится внутренняя переменная процедуры, которая используются только в ней и ей присваивается значение фактического параметра. Другим видом параметров являются параметры-переменные. Они позволяют указать переменные, в которых процедура будет сохранять результаты своей работы. Наша процедура все еще далека от идеала. Она всегда заносит результат в переменную y и жестко привязана к ней. Рассмотрим третий вариант:
…
Procedure digiSum2(i:integer; var sum: integer); { поиск суммы цифр целого числа (sum:= сумма цифр i) }

begin
sum:=0; i:=abs(i);

while (i>0) do begin


sum:=sum + i mod 10; i:=i div 10;

end;

end;
begin
read(x); digiSum2(x,x); digiSum2(x,x); writeln(y)

end.
Подстановка параметра-переменной происходит так, как если бы каждое вхождение sum в разделе операторов процедуры мы заменили на фактический параметр x.
Мы видим, что нет нужды в переменной y, процедура может сохранять результат в любой свободной переменной, даже в x. В заголовке процедуры параметры-значения отличаются от параметров-переменных тем, что перед параметрами-переменными стоит служебное слово var. 

При вызове процедуры в списке фактических параметров на месте параметра-значения можно указывать выражение совместимого по присваиванию типа. На месте параметра-переменной можно указывать только переменную совместимого типа.


Рассмотрим окончание записи заголовка процедуры:

<список формальных параметров>::= <элемент списка параметров> {; <элемент списка параметров>}
<элемент списка параметров>::= [var] <имя формального параметра> {, <имя формального параметра>}: <имя типа>

Важно! В списке формальных параметров не может использоваться задание типа, только имя типа. ​

Параметр может иметь производный тип. Объявляя параметр-значение производного типа, нужно иметь в виду, что в процедуре будет использована память под внутреннюю переменную того же типа (например, массив) и туда будет скопировано значение фактического параметра, все это может занять дополнительное время и дополнительную память. Поэтому иногда выгоднее объявить параметр производного типа параметром-переменной, даже если процедура не меняет его значения.

Заголовок процедуры может содержать директиву forward, в тех случаях, когда программист хочет дать опережающее описание. Пример:
Procedure digiSum3(i:integer; var sum: integer); forward; 
{ поиск суммы цифр целого числа (sum:= сумма цифр i) }
…
Procedure digiSum3

begin
sum:=0; i:=abs(i); …

Такое описание позволяет использовать в программе процедуру раньше, чем она была описана. Это бывает удобно в некоторых специальных случаях.

Вернемся к задаче поиска НОД четырех чисел.

program NOD4(input, output);

var a,b,c,d:integer; 
procedure NOD2(a,b:integer; var c:integer);

begin 
while (a <> 0) and (b<>0) do begin



if a<b then begin




b:= b mod a


end else begin




a:=a mod b


end;


end;


if a=0 then begin



c:=b;


end else begin



c:=a

end;

end;

begin 
read(a,b,c,d);


NOD1(c,d,c); {c:=нод(c,d)}


NOD1(b,c,b); {b:=нод(b,нод(c,d))}


NOD1(a,b,a); {a:=нод(a,нод(b,нод(c,d)))}


writeln(a)

end.

В примере мы использовали одни и те же имена для обозначения переменных программы и параметров процедуры. Это допустимо благодаря принципу локализации: 

Все имена вводимые в какой-либо процедуре имеют силу только в пределах этой процедуры. Имя из родительского блока или других объемлющих блоков, совпадающее с внутренним именем процедуры «заслоняется» внутренним именем и не может быть использовано внутри процедуры в старом смысле.


В примере, формальные параметры a, b, c «заслоняют» одноименные переменные программы, поэтому при выполнении процедуры значения переменных программы не меняются. Процедура меняет лишь значение фактического параметра-переменной.   

Рассмотрим какие имена могут встретиться внутри процедуры:
Глобальные имена ​– это имена описанные в блоке программы, не совпадающие с именами формальных параметров и внутренних переменных процедуры. Для процедуры digiSum​ x и y ​– глобальные имена. Изменение значений глобальных переменных в процедуре чревато ошибками, поскольку можно забыть про него, ведь глобальные имена отсутствуют в списке параметров. Поэтому рекомендуется использовать параметры-значения, явно указывая, что процедура меняет значения переменных.
Формальные параметры – это имена из списка формальных параметров. Каждому формальному параметру-значению соответствует переменная, создаваемая при вызове процедуры и уничтожаемая по выходу из процедуры. Если глобальное имя совпадает с именем формального параметра, оно перестает быть видимым внутри блока процедуры. Такое имя везде внутри процедуры означает формальный параметр.
Локальные имена – это имена, описанные в блоке процедуры. Они не могут совпадать с именами формальных параметров. Каждая локальная переменная существует лишь пока происходит выполнение раздела операторов процедуры. По окончании переменная уничтожается, её значение теряется.

Иногда непросто определиться какое имя внутри программы что означает. Пример:

Определить, что будет выдано на печать:

program Пример(output);

var
a,b,c,d: integer;

procedure P(x:integer; var a:integer);

var c:integer;

begin 
x:=5; a:=6; c:=7; d:=8; 
writeln( a,b,c,d) 
end;

begin 
a:=1; b:=2; c:=3; d:=4; P(a,b); 
writeln(a,b,c,d) 
end.
Чтобы разобраться нарисуем таблицу.

	A
	B
	c
	d
	Px
	Pa
	Pc

	
	
	
	
	
	
	


Заголовками первых 4-х столбцов являются имена глобальных переменных, следующие 2 столбца – формальные параметры, 7-ой столбец – локальная переменная (мы специально приписали перед формальными параметрами и локальной переменной P, чтобы отличить их от глобальных имен). Теперь будем моделировать выполнение программы и заполнять таблицу. После выполнения первых 4-х операторов присваивания таблица примет вид:

	A
	B
	c
	d
	Px
	Pa
	Pc

	1
	2
	3
	4
	
	
	


При вызове процедуры произойдет подстановка параметров. В параметр-значение будет скопировано значение фактического параметра, а формальный параметр будет объявлен синонимом фактического:

	A
	b
	c
	d
	Px
	Pa
	Pc

	1
	2
	3
	4
	1
	b
	


Начинается выполнение раздела операторов процедуры. Поскольку имена a и с используются как имена формального параметра и локальной переменной и как имена глобальных объектов, то глобальные a и c внутри процедуры не видны, работа идет с внутренними именами. После выполнения присваиваний таблица примет вид:
	A
	b
	c
	d
	Px
	Pa
	Pc

	1
	2
6
	3
	4
8
	1
5
	b
	7


Обратите внимание, при присваивании локальному a, меняется глобальное b, поскольку b – фактический параметр-переменная. Оператор writeln внутри процедуры напечатает: 5678; оператор writeln внутри раздела операторов программы напечатает 1638.
Функции


Зачастую основным результатом процедуры является найденное значение простого типа. В таких случаях лучше оформлять решение подзадачи не как процедуру, а как функцию. 
<описание функции>::= <заголовок функции> ; (<блок> | <директива>) | 
function <имя> ; <блок>
<заголовок процедуры> ::= function <имя> [( <список формальных параметров> )] : <тип результата>
<тип результата> ::= <имя простого типа>

Вызов функции выглядит также как вызов процедуры, но является не оператором, а выражением, которое имеет тип результата функции.


Перепишем поиск суммы цифр в виде функции:
function digiSum4(i:integer): integer; { поиск суммы цифр целого числа i }
var sum:integer;
begin
sum:=0; i:=abs(i);

while (i>0) do begin


sum:=sum + i mod 10; i:=i div 10;

end;
digiSum4:=sum
end;
Обратите внимание на последний оператор, он явным образом указывает значение, возвращаемое в качестве результата. Такой оператор обязательно должен выполниться внутри функции (не обязательно один раз) в любом случае, иначе произойдет ошибка, так как функция обязательно должна возвратить результат. Имя функции нельзя использовать как временную переменную (именно поэтому мы накапливаем результат в локальной переменной sum). Имя функции можно использовать лишь в левой части оператора присваивания для указания значения результата и при обращении к функции (со списком фактических параметров). Например, мы можем определить функцию поиска суммы цифр с помощью обращения к самой себе. Заметим такой факт, что сумма цифр числа S(x)=x mod 10 + S(x div 10). Используя это опишем:
function digiSum5(i:integer): integer; { поиск суммы цифр целого числа i рекурсивно}
begin
i:=abs(i);

if i<10 then begin



digiSum5:=i
{нерекурсивная ветвь}


end else begin
digiSum5:=i mod 10 + digiSum5(i div 10)
{рекурсивная ветвь}

end
end;
Рассмотрим, как будет вычисляться digiSum5(123). При первом вызове сработает рекурсивная ветвь по которой digiSum5:=3 + digiSum5(12), и произойдет рекурсивный вызов. При вычислении digiSum5(12) снова сработает рекурсивная ветвь digiSum5:=2 + digiSum5(1). Произойдет второй рекурсивный вызов. При вычислении digiSum5(1) сработает нерекурсивная ветвь, произойдет возврат и результат (1) подставится вместо второго слагаемого при вычислении digiSum5(12) : digiSum5:=2 + 1. Сумма будет найдена и снова произойдет возврат к первоначальному вызову. Результат (3) подставится вместо второго слагаемого при вычислении digiSum5(123) : digiSum5:=3 + 3, после чего будет найден окончательный результат: 6.

Заметим, что без особой нужды не следует использовать рекурсию. Доказано, что любой цикл может быть реализован рекурсивно (обратное тоже верно, но реализуется труднее). Рекурсия вызывает в программе побочные расходы на хранение значений параметров и локальных переменных. Если произойдет определенное количество рекурсивных вызовов (и не будет сделано ни одного возврата) исчерпается память и программа остановится с ошибкой. Именно поэтому реализация в виде цикла является более предпочтительной, чем рекурсивная реализация.
Примеры использования рекурсии

Подходящим случаем использования рекурсии является задача о ханойских башнях. type стержни=1..3;

Procedure Move(i: integer; m,n:стержни); {перенос i дисков со стержня m на стержень n}
var s:стержни;
begin
if i=1 then begin



writeln(m, '->', n)


end else begin



s:=6-m-n;



move(i-1,m,s);



writeln(m, '->', n)



move(i-1,s,n);


end;

end;

Другой подходящий случай: генерация всех возможных перестановок из чисел 1..n (n – константа). Будем хранить перестановки в массиве a.
Program Перестановки(input, output);
Const n=6;
Var a:array [1..n] of integer; t,i:integer;
Procedure swap(i,j:integer);

Var k:integer;

Begin k:=a[i]; a[i]:=a[j]; a[j]:=k

End;

Procedure generate;

Var i,j:integer;

Begin
if t=n then begin



For i:=1 to n do begin




Write(a[i])



End;



Writeln


End else begin



For j:=t+1 to n do begin




Swap(t+1,j);




t:=t+1;




generate;




t:=t-1;




swap(t+1,j)



end;


end;

end;

begin
for i:=1 to n do begin



a[i]:=i

end;


t:=0;


generate

end.

Последнее замечание. Вспомним о пошаговой детализации. В этом процессе задача разбивается на подзадачи и решение записывается как оператор псевдокода. Очевидно что при пошаговой детализации удобно использовать процедуры и функции, которые как раз и предназначены для реализации подзадач в рамках основной задачи. Таким образом, псевдооператоры можно заменять вызовами процедур или операторами присваивания, в правой части которых стоят вызовы функций.

Еще один пример, задача расположения n ферзей на доске размером n х n так, чтобы никакие два ферзя не «били» друг друга.
program Queens(input,output); {расстановка ферзей}
const n=8; {количество ферзей, заметим, что для n=3 или 2 решения нет}

var i,j:integer; d: array [1..n,1..n] of char; s:longint;

function check(b:integer):boolean; 
{проверка, можно ли поставить ферзя в клетку i-ой строки и b-ого столбца}

var k,m:integer; t:boolean;

begin   k:=1; t:=true;

        while (k<=i) and t do begin

                t:=t and (d[k,b]<>'Q'); {проверка отсутствия ферзей на вертикали b}

                k:=k+1;

        end;

        k:=1;

        while (k<=n) and t do begin {проверка отсутствия ферзей на диагоналях}

                m:=i-b+k+1;

                if (m>0) and (m<=i) then begin

                        t:=t and (d[m,k]<>'Q');

                end;

                m:=i+b-k+1;

                if (m>0) and (m<=i) then begin

                        t:=t and (d[m,k]<>'Q');

                end;

                k:=k+1;

        end;

        check:=t

end;

procedure gen; {рекурсивная процедура, осуществляющая все возможные расстановки}

var a,b,c:integer;ch:char;

begin

        if i=n then begin

                for a:=1 to n do begin

                        for c:=1 to n do begin

                                write(d[a,c]); {печать завершенной расстановки}

                        end;

                        writeln;

                end;

                writeln; s:=s+1; {увеличение количества успешных расстановок}

        end else begin

                for b:=1 to n do begin
{проверяем все позиции в i+1-ой строке}

                  if check(b) then begin
{можно ли туда поставить очередного ферзя}

                        d[i+1,b]:='Q';

                        i:=i+1;

                        gen;

                        i:=i-1;

                        d[i+1,b]:='+';

{+ означает пустую клетку}

                  end;

                end

        end

end;

begin




{раздел операторов программы}
        for i:=1 to n do begin

                for j:=1 to n do begin

                        d[i,j]:='+';

{заполняем массив-доску пустыми клетками}

                end
        end;

        i:=0;s:=0;

        gen;



{генерируем расстановки ферзей}

        writeln(s);


{выводим ответ}

end.

